Mechanics 2341, PS 4 2018

```
FS = FullSimplify;
```

Problem 1

Find the one - dimensional particle motion in the Poschl - Teller potential U[x, a, V] = -V/Cosh[a x]^2, representing an atomic well.

```
Clear[x, a, A, m, En]
$Assumptions = {{a, V, En, x} ∈ Reals};
```

Poschl-Teller's potential as a function of x, a and V

```
U[x_, a_, V_] = -V / Cosh[a x]^2
```

$-V\,\mathrm{Sech}[a\,x]^2$

```
Plot[U[x, 1, 1], {x, -3, 3}, PlotRange → {-1.5, 0.5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {x, U}}]
```



To analyze the motion it is convenient to introduce a generalized coordinate y: y=Sinh[a x]. In terms of the generalized coordinate the kinetic and potential energy take the form

```
x = 1 / a ArcSinh[y];
Ekin = m / 2 D[x, y]^2 (dy / dt)^2
```

$$\frac{dy^2\,m}{2\,a^2\,dt^2\,(1+y^2)}$$

```
Epot = U[x, a, V]
```

$$-\frac{V}{1+y^2}$$

Here is the plot of U as a function of y

```
Plot[U[x /. {a → 1}, 1, 1], {y, -8, 8}, PlotRange → {-1.5, 0.5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {y, U}}]
```



The total energy is therefore

```
EE = Ekin + Epot
```

$$\frac{dy^2\, m}{2\, a^2\, dt^2\, \left(1+y^2\right)}-\frac{V}{1+y^2}$$

Solving the equation En= Ekin+Epot for dt, we get

```
Assuming[a > 0, Simplify[Solve[En == EE, dt]]]
```

$$\left\{\left\{dt\to -\frac{dy\,\sqrt{m}}{\sqrt{2}\,a\,\sqrt{En+V+En\,y^2}}\right\},\left\{dt\to \frac{dy\,\sqrt{m}}{\sqrt{2}\,a\,\sqrt{En+V+En\,y^2}}\right\}\right\}$$

Thus, t as a function of y is given by the integral

$$t = \text{Integrate}\left[\frac{\sqrt{m}}{\sqrt{2}\; a \sqrt{En + V + En\, y^2}},\ y\right]$$

Case I: Finite motion, -V < En < 0

```
Plot[{U[x /. {a → 1}, 1, 1], -0.6}, {y, -8, 8}, PlotRange → {-1.5, 0.5},
 PlotStyle → {{Thickness[0.008], RGBColor[0, 0, 1]},
   {Thickness[0.008], RGBColor[1, 0, 0]}}, {AxesLabel → {y, U}}]
```



Turning points; En is the energy

```
Assuming[{En < 0, V > 0, V > -En}, FullSimplify[Solve[U[x, a, V] == En, y]]]
```

$$\left\{\left\{y \to \frac{\sqrt{-En\,(En+V)}}{En}\right\},\ \left\{y \to \sqrt{-\frac{En+V}{En}}\right\}\right\}$$

```
Ymin = -Sqrt[-1 - V/En];
Ymax = -Ymin;
```

Period of oscillations

```
T = Assuming[{En < 0, a > 0, V > 0, V > -En},
  FullSimplify[(2 m)^(1/2) Integrate[1/(a Sqrt[En + V + En y^2]), {y, Ymin, Ymax}]]]
```

$$\frac{\sqrt{2}\sqrt{-\frac{m}{En}}\,\pi}{a}$$

Thus

$$T = \frac{\sqrt{2}\,\sqrt{-\frac{m}{En}}\,\pi}{a};$$

As En approaches - V, we reproduce the usual period of harmonic oscillations. As En approaches 0, the period goes to infinity.

y and x as functions of t

En = -Em;

$$\text{Assuming}\Big[\Big\{Em > 0,\ V > 0,\ V > -En,\ y > 0,\ y < \sqrt{-(En + V)/En}\Big\},$$
$$\text{FS}\Big[\text{Integrate}\Big[\frac{\sqrt{m}}{\sqrt{2}\ a\ \sqrt{En + V + En\ z^2}},\ \{z, 0, y\}\Big]\Big]\Big]$$

$$\frac{\sqrt{\frac{m}{Em}}\ \text{ArcSin}\Big[\sqrt{\frac{Em}{-Em+V}}\ y\Big]}{\sqrt{2}\ a}$$

$$\text{Assuming}\Big[\Big\{a > 0,\ t > 0,\ m > 0,\ Em > 0,\ V > 0,\ V > -En,\ y > 0,\ y < \sqrt{-(En + V)/En}\Big\},$$
$$\text{Solve}\Big[\frac{\sqrt{\frac{m}{Em}}\ \text{ArcSin}\Big[\sqrt{\frac{Em}{-Em+V}}\ y\Big]}{\sqrt{2}\ a} == t,\ y\Big]\Big]$$

$$\Bigg\{\Bigg\{y \to \text{ConditionalExpression}\Bigg[\frac{\text{Sin}\Big[\frac{\sqrt{2}\ a\ t}{\sqrt{\frac{m}{Em}}}\Big]}{\sqrt{-\frac{Em}{Em-V}}},\ \left(\sqrt{2}\ \text{Re}\Big[\frac{a\ t}{\sqrt{\frac{m}{Em}}}\Big] == -\frac{\pi}{2}\ \&\&\ \sqrt{2}\ \text{Im}\Big[\frac{a\ t}{\sqrt{\frac{m}{Em}}}\Big] \geq 0\right)\ ||$$
$$-\frac{\pi}{2} < \sqrt{2}\ \text{Re}\Big[\frac{a\ t}{\sqrt{\frac{m}{Em}}}\Big] < \frac{\pi}{2}\ ||\ \left(\sqrt{2}\ \text{Re}\Big[\frac{a\ t}{\sqrt{\frac{m}{Em}}}\Big] == \frac{\pi}{2}\ \&\&\ \sqrt{2}\ \text{Im}\Big[\frac{a\ t}{\sqrt{\frac{m}{Em}}}\Big] \leq 0\right)\Bigg]\Bigg\}\Bigg\}$$

$$Y[t\_] = \frac{\text{Sin}\Big[\frac{\sqrt{2}\ a\ t}{\sqrt{\frac{m}{Em}}}\Big]}{\sqrt{-\frac{Em}{Em-V}}};$$

```
Plot[(Y[t] /. {Em → 1/2, a → 1, m → 1, V -> 1}), {t, 0, 20 Pi}]
```




```
Plot[(Y[t] /. {Em → 1/10, a → 1, m → 1, V -> 1}), {t, 0, 20 Pi}]
```




```
Plot[(Y[t] /. {Em → 1/100, a → 1, m → 1, V -> 1}), {t, 0, 20 Pi}]
```



In terms of x coordinate one gets

```
X[t_] = Assuming[{Em > 0, V > 0, a > 0, m > 0}, FS[x /. y → Y[t]]]
```

$$\frac{\mathrm{ArcSinh}\left[\frac{\mathrm{Sin}\left[\sqrt{2}\ a\ \sqrt{\frac{\mathrm{Em}}{\mathrm{m}}}\ t\right]}{\sqrt{\frac{\mathrm{Em}}{-\mathrm{Em}+\mathrm{V}}}}\right]}{a}$$

```
Plot[(X[t] /. {Em → 1/2, a → 1, m → 1, V -> 1}), {t, 0, 20 Pi}]
```




```
Plot[(X[t] /. {Em → 1/10, a → 1, m → 1, V -> 1}), {t, 0, 20 Pi}, PlotRange → All]
```



```
Plot[(X[t] /. {Em → 1/100, a → 1, m → 1, V -> 1}), {t, 0, 20 Pi}]
```



Case II: Infinite motion, En > 0

```
Plot[{U[x /. {a → 1}, 1, 1], 0.6}, {y, -8, 8},
 PlotRange → {-1.5, 1}, {AxesLabel → {y, U}}, PlotStyle →
  {{Thickness[0.008], RGBColor[0, 0, 1]}, {Thickness[0.008], RGBColor[1, 0, 0]}}]
```



There is no turning point. Let's find the time delay, that is the difference between the (infinite) time spent by the interacting particle and the time spent by a non-interacting particle of the same energy to get from -infinity to +infinity

```
Clear[En]
```

```
Tdelay = (2 m)^(1/2) Assuming[{En > 0, V > 0}, Integrate[
   1/(a Sqrt[En + V + En y^2]) - 1/En^(1/2) 1/(a (1 + y^2)^(1/2)), {y, 0, Infinity}]]
```

$$\frac{\sqrt{m}\ \mathrm{Log}\left[\frac{\mathrm{En}}{\mathrm{En}+V}\right]}{\sqrt{2}\ a\ \sqrt{\mathrm{En}}}$$

$$\text{Tdelay} = \frac{\sqrt{m}\ \text{Log}\left[\frac{\text{En}}{\text{En}+V}\right]}{\sqrt{2}\ a\ \sqrt{\text{En}}};$$

```
Series[Tdelay, {En, 0, 0}] // Normal // Expand
```

$$\frac{\sqrt{m}\ \text{Log}[\text{En}]}{\sqrt{2}\ a\ \sqrt{\text{En}}} + \frac{\sqrt{m}\ \text{Log}\left[\frac{1}{V}\right]}{\sqrt{2}\ a\ \sqrt{\text{En}}}$$

```
Series[Tdelay, {En, Infinity, 3}] // Normal // Expand
```

$$-\frac{\left(\frac{1}{\text{En}}\right)^{3/2}\sqrt{m}\ V}{\sqrt{2}\ a} + \frac{\left(\frac{1}{\text{En}}\right)^{5/2}\sqrt{m}\ V^2}{2\sqrt{2}\ a}$$

Since $\text{Log}\left[\frac{\text{En}}{\text{En}+V}\right] < 0$ the time delay is negative, so the interacting particle gets to + infinity faster than the non - interacting one. For slowly moving particle the time delay goes to infinity. If the energy goes to infinity the time delay goes to 0, as one should expect.

y and x as functions of t

$$\text{Assuming}\left[\{\text{En} > 0, V > 0, y > 0\}, \text{FS}\left[\text{Integrate}\left[\frac{\sqrt{m}}{\sqrt{2}\ a\ \sqrt{\text{En} + V + \text{En}\ z^2}}, \{z, 0, y\}\right]\right]\right]$$

$$\frac{\text{ArcSinh}\left[\sqrt{\frac{\text{En}}{\text{En}+V}}\ y\right]}{\sqrt{2}\ a\ \sqrt{\frac{\text{En}}{m}}}$$

$$\text{Assuming}\left[\{a > 0, t > 0, m > 0, \text{En} > 0, V > 0, y > 0\}, \text{FS}\left[\text{Solve}\left[\frac{\text{ArcSinh}\left[\sqrt{\frac{\text{En}}{\text{En}+V}}\ y\right]}{\sqrt{2}\ a\ \sqrt{\frac{\text{En}}{m}}} == t, y\right]\right]\right]$$

$$\left\{\left\{y \to \sqrt{\frac{\text{En}+V}{\text{En}}}\ \text{Sinh}\left[\sqrt{2}\ a\ \sqrt{\frac{\text{En}}{m}}\ t\right]\right\}\right\}$$

$$\text{Y[t\_]} = y\ /.\ \left\{y \to \sqrt{\frac{\text{En}+V}{\text{En}}}\ \text{Sinh}\left[\sqrt{2}\ a\ \sqrt{\frac{\text{En}}{m}}\ t\right]\right\}$$

$$\sqrt{\frac{\text{En}+V}{\text{En}}}\ \text{Sinh}\left[\sqrt{2}\ a\ \sqrt{\frac{\text{En}}{m}}\ t\right]$$

```
Plot[{(Y[t] /. {En → 1/100, a → 1, m → 1, V -> 1}),
  (Y[t] /. {En → 1/10, a → 1, m → 1, V -> 1}),
  (Y[t] /. {En → 1, a → 1, m → 1, V -> 1})}, {t, -Pi, Pi}]
```




```
Plot[(Y[t] /. {En → 1/100, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```




```
Plot[(Y[t] /. {En → 1/10, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



```
Plot[(Y[t] /. {En → 1, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



In terms of x coordinate one gets

```
X[t_] = Assuming[{En > 0, V > 0, a > 0, m > 0}, FS[x /. y → Y[t]]]
```

$$\frac{\text{ArcSinh}\left[\sqrt{\frac{\text{En}+\text{V}}{\text{En}}}\ \text{Sinh}\left[\sqrt{2}\ \text{a}\ \sqrt{\frac{\text{En}}{\text{m}}}\ \text{t}\right]\right]}{\text{a}}$$

```
Plot[{(X[t] /. {En → 1/100, a → 1, m → 1, V -> 1}),
  (X[t] /. {En → 1/10, a → 1, m → 1, V -> 1}),
  (X[t] /. {En → 1, a → 1, m → 1, V -> 1})}, {t, -Pi, Pi}]
```



```
Plot[(X[t] /. {En → 1/100, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



```
Plot[(X[t] /. {En → 1/10, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



```
Plot[(X[t] /. {En → 1, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



Problem 2

A particle is moving with the energy E > V in the Poschl - Teller potential U[x, a, V] = V/Cosh[a x]^2, representing an atomic barrier.Find the time delay in its motion from minus to plus infinity compared to a free moving particle.

```
Clear[x, a, V, m, En]
```

```
$Assumptions = {{a, V, En, x} ∈ Reals};
```

Poschl-Teller's potential as a function of x, a and V

```
U[x_, a_, V_] = V / Cosh[a x] ^2
```

$V \, \mathrm{Sech}[a\, x]^2$

```
Plot[U[x, 1, 1], {x, -3, 3}, PlotRange → {-0.5, 1.5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {x, U}}]
```



To analyze the motion it is convenient to introduce a generalized coordinate y: y=Sinh[a x]. In terms of the generalized coordinate the kinetic and potential energy take the form

```
x = 1 / a ArcSinh[y];
```

```
Ekin = m / 2 D[x, y] ^2 (dy / dt) ^2
```

$$\frac{dy^2\, m}{2\, a^2\, dt^2\, \left(1 + y^2\right)}$$

```
Epot = U[x, a, V]
```

$$\frac{V}{1 + y^2}$$

Here is the plot of U as a function of y

```
Plot[U[x /. {a → 1}, 1, 1], {y, -8, 8}, PlotRange → {-0.5, 1.5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {y, U}}]
```



The total energy is therefore

```
EE = Ekin + Epot
```

$$\frac{dy^2\, m}{2\, a^2\, dt^2\, \left(1+y^2\right)} + \frac{V}{1+y^2}$$

Solving the equation En= Ekin+Epot for dt, we get

```
Assuming[a > 0, Simplify[Solve[En == EE, dt]]]
```

$$\left\{\left\{dt \to -\frac{dy\,\sqrt{m}}{\sqrt{2}\, a\, \sqrt{En - V + En\, y^2}}\right\}, \left\{dt \to \frac{dy\,\sqrt{m}}{\sqrt{2}\, a\, \sqrt{En - V + En\, y^2}}\right\}\right\}$$

Thus, t as a function of y is given by the integral

$$t = \text{Integrate}\left[\frac{\sqrt{m}}{\sqrt{2}\, a \sqrt{En - V + En\, y^2}}, y\right]$$

Infinite motion, En > V

```
Plot[{U[x /. {a → 1}, 1, 1], 1.6}, {y, -8, 8},
 PlotRange → {-0.5, 2}, {AxesLabel → {y, U}}, PlotStyle →
  {{Thickness[0.008], RGBColor[0, 0, 1]}, {Thickness[0.008], RGBColor[1, 0, 0]}}]
```



There is no turning point. Let's find the time delay, that is the difference between the (infinite) time spent by the interacting particle and the time spent by a non-interacting particle of the same energy to get from -infinity to +infinity

```
Tdelay = (2 m) ^ (1/2) Assuming[{En > V, V > 0}, Integrate[
    1/(a Sqrt[En - V + En y^2]) - 1/En^(1/2) 1/(a (1 + y^2)^(1/2)), {y, 0, Infinity}]]
```

$$\frac{\sqrt{m}\, \text{Log}\left[\frac{En}{En-V}\right]}{\sqrt{2}\, a \sqrt{En}}$$

$$\text{Tdelay} = \frac{\sqrt{m}\, \text{Log}\left[\frac{En}{En-V}\right]}{\sqrt{2}\, a \sqrt{En}};$$

```
Series[Tdelay, {En, V, 0}] // Normal // Expand
```

$$-\frac{\sqrt{m}\, \text{Log}[En - V]}{\sqrt{2}\, a \sqrt{V}} + \frac{\sqrt{m}\, \text{Log}[V]}{\sqrt{2}\, a \sqrt{V}}$$

```
Series[Tdelay, {En, Infinity, 3}] // Normal // Expand
```

$$\frac{\left(\frac{1}{En}\right)^{3/2}\sqrt{m}\ V}{\sqrt{2}\ a}+\frac{\left(\frac{1}{En}\right)^{5/2}\sqrt{m}\ V^2}{2\sqrt{2}\ a}$$

Since $\text{Log}\left[\frac{En}{En - V}\right] > 0$ the time delay is positive, so the non – interacting particle gets to + infinity faster than the interacting one. When the energy goes to infinity the time delay goes to 0, as one should expect.

y and x as functions of t

$$\text{Assuming}\left[\{En > V, V > 0, y > 0\}, \text{FS}\left[\text{Integrate}\left[\frac{\sqrt{m}}{\sqrt{2}\ a\sqrt{En - V + En\ z^2}}, \{z, 0, y\}\right]\right]\right]$$

$$\frac{\text{ArcSinh}\left[\sqrt{\frac{En}{En-V}}\ y\right]}{\sqrt{2}\ a\sqrt{\frac{En}{m}}}$$

$$\text{Assuming}\left[\{a > 0, t > 0, m > 0, En > V, V > 0, y > 0\}, \text{FS}\left[\text{Solve}\left[\frac{\text{ArcSinh}\left[\sqrt{\frac{En}{En-V}}\ y\right]}{\sqrt{2}\ a\sqrt{\frac{En}{m}}} == t, y\right]\right]\right]$$

$$\left\{\left\{y \to \sqrt{1 - \frac{V}{En}}\ \text{Sinh}\left[\sqrt{2}\ a\sqrt{\frac{En}{m}}\ t\right]\right\}\right\}$$

$$\text{Y[t\_] = y /. }\left\{y \to \sqrt{1 - \frac{V}{En}}\ \text{Sinh}\left[\sqrt{2}\ a\sqrt{\frac{En}{m}}\ t\right]\right\}$$

$$\sqrt{1 - \frac{V}{En}}\ \text{Sinh}\left[\sqrt{2}\ a\sqrt{\frac{En}{m}}\ t\right]$$

```
Plot[{(Y[t] /. {En → 1 + 1/100, a → 1, m → 1, V -> 1}),
  (Y[t] /. {En → 1 + 1/10, a → 1, m → 1, V -> 1}),
  (Y[t] /. {En → 2, a → 1, m → 1, V -> 1})}, {t, -Pi, Pi}]
```




```
Plot[(Y[t] /. {En → 1 + 1/100, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```




```
Plot[(Y[t] /. {En → 1 + 1/10, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



```
Plot[(Y[t] /. {En → 2, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



In terms of x coordinate one gets

```
X[t_] = Assuming[{En > V, V > 0, a > 0, m > 0}, FS[x /. y → Y[t]]]
```

$$\frac{\text{ArcSinh}\left[\sqrt{1-\frac{V}{En}}\ \text{Sinh}\left[\sqrt{2}\ a\ \sqrt{\frac{En}{m}}\ t\right]\right]}{a}$$

```
Plot[{(X[t] /. {En → 1 + 1/100, a → 1, m → 1, V -> 1}),
  (X[t] /. {En → 1 + 1/10, a → 1, m → 1, V -> 1}),
  (X[t] /. {En → 2, a → 1, m → 1, V -> 1})}, {t, -Pi, Pi}]
```



```
Plot[(X[t] /. {En → 1 + 1/100, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



```
Plot[(X[t] /. {En → 1 + 1/10, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



```
Plot[(X[t] /. {En → 2, a → 1, m → 1, V -> 1}), {t, -Pi, Pi}]
```



Problem 3

Find the one - dimensional particle motion in the trigonometric potential U[x, a, V] = V Tan[a x]^2

```
Clear[x, a, V, m, En]
$Assumptions = {{a, V, En, x} ∈ Reals};
```

Trigonometric potential as a function of x, a and V

```
U[x_, a_, V_] = V Tan[a x]^2
```

$V \, \mathrm{Tan}[a\, x]^2$

```
Plot[U[x, 1, 1], {x, -6, 6}, PlotRange → {-0.5, 7.5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {x, U}}]
```



To analyze the motion it is convenient to introduce a generalized coordinate y: y=Sin[a x]. The range of y is -1 < y < 1. In terms of the generalized coordinate the kinetic and potential energy take the form

```
x = 1 / a ArcSin[y];
Ekin = m / 2 D[x, y]^2 (dy / dt)^2
```

$$\frac{dy^2\, m}{2\, a^2\, dt^2\, \left(1 - y^2\right)}$$

```
Epot = U[x, a, V]
```

$$\frac{V\, y^2}{1 - y^2}$$

Here is the plot of U as a function of y

```
Plot[U[x /. {a → 1}, 1, 1], {y, -1, 1}, PlotRange → {-0.5, 8.5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {y, U}}]
```



The total energy is therefore

```
EE = Ekin + Epot
```

$$\frac{dy^2\, m}{2\, a^2\, dt^2\, \left(1 - y^2\right)} + \frac{V\, y^2}{1 - y^2}$$

Solving the equation En= Ekin+Epot for dt, we get

```
Assuming[a > 0, Simplify[Solve[En == EE, dt]]]
```

$$\left\{\left\{dt \to -\frac{dy\, \sqrt{m}}{\sqrt{2}\, a\, \sqrt{En - En\, y^2 - V\, y^2}}\right\}, \left\{dt \to \frac{dy\, \sqrt{m}}{\sqrt{2}\, a\, \sqrt{En - En\, y^2 - V\, y^2}}\right\}\right\}$$

Thus, t as a function of y is given by the integral

$$t = \text{Integrate}\left[\frac{\sqrt{m}}{\sqrt{2}\, a\, \sqrt{En - En\, y^2 - V\, y^2}}, y\right]$$

which can be explicitly computed, see below

The only case: Finite motion, En > 0

```
Plot[{U[x /. {a → 1}, 1, 1], 2.6}, {y, -1, 1}, PlotRange → {-1.5, 8.5},
 PlotStyle → {{Thickness[0.008], RGBColor[0, 0, 1]},
   {Thickness[0.008], RGBColor[1, 0, 0]}}, {AxesLabel → {y, U}}]
```



Turning points; En is the energy

```
Assuming[{En > 0, V > 0}, FullSimplify[Solve[U[x, a, V] == En, y]]]
```

$$\left\{\left\{y \to -\sqrt{\frac{En}{En + V}}\right\}, \left\{y \to \sqrt{\frac{En}{En + V}}\right\}\right\}$$

$$Ymin = -\sqrt{\frac{En}{En + V}};$$

```
Ymax = -Ymin;
```

Period of oscillations

$$T = Assuming\left[\{En > 0, a > 0, V > 0\},\ FullSimplify\left[(2\,m)^{\wedge}(1/2)\ Integrate\left[1 \Big/ \left(a\sqrt{En - En\,y^2 - V\,y^2}\right), \{y, Ymin, Ymax\}\right]\right]\right]$$

$$\frac{\sqrt{2}\,\pi\sqrt{\frac{m}{En+V}}}{a}$$

Thus

$$T = \frac{\sqrt{2}\,\pi\sqrt{\frac{m}{En+V}}}{a};$$

```
Series[T, {En, 0, 1}] // FS // Normal
```

$$\frac{\sqrt{2}\,\pi\sqrt{\frac{m}{V}}}{a} - \frac{En\,\pi\sqrt{\frac{m}{V}}}{\sqrt{2}\,a\,V}$$

```
Series[T, {En, Infinity, 1}] // FS // Normal
```

$$\frac{\sqrt{2}\sqrt{\frac{1}{En}}\sqrt{m}\,\pi}{a}$$

As En approaches 0, we reproduce the usual period of harmonic oscillations. As En approaches infinity, the period goes to 0 as one should expect because the velocity of the particle goes to infinity and the range of x is finite; the particle is trapped in the box of size Pi/a .

y and x as functions of t

```
Assuming[{En > 0, y > 0, y < (En/(En+V))^(1/2), V > 0},
 FS[Integrate[Sqrt[m]/(Sqrt[2] a Sqrt[En - En z^2 - V z^2]), {z, 0, y}]]]
```

$$\frac{\sqrt{\frac{m}{En+V}}\,\mathrm{ArcSin}\left[\sqrt{\frac{En+V}{En}}\,y\right]}{\sqrt{2}\,a}$$

Inverting the function one finds y as a function of t

```
Assuming[{En > 0, y > 0, y < (En/(En+V))^(1/2), V > 0, a > 0, m > 0, t > 0},
 FS[Solve[Sqrt[m/(En+V)] ArcSin[Sqrt[(En+V)/En] y]/(Sqrt[2] a) == t, y]]]
```

$$\left\{\left\{y \to \mathrm{ConditionalExpression}\left[\frac{En\,\mathrm{Sin}\left[\sqrt{2}\,a\,t\sqrt{\frac{En+V}{m}}\right]}{\sqrt{En\,(En+V)}},\ 2\sqrt{2}\,a\,t\sqrt{\frac{En+V}{m}} \le \pi\right]\right\}\right\}$$

$$Y[t\_] = \frac{En\,\mathrm{Sin}\left[\sqrt{2}\,a\,t\sqrt{\frac{En+V}{m}}\right]}{\sqrt{En\,(En+V)}};$$

We see from the solution that the period of oscillations is $T = \frac{\sqrt{2}\,\pi\sqrt{\frac{m}{En+V}}}{a}$.

Here is the plot of the solution

```
Plot[(Y[t] /. {En → 1, a → 1, m → 1, V -> 1}), {t, 0, Pi}]
```




```
Plot[(Y[t] /. {En → 10, a → 1, m → 1, V -> 1}), {t, 0, Pi}]
```




```
Plot[(Y[t] /. {En → 100, a → 1, m → 1, V -> 1}), {t, 0, Pi}]
```



In terms of x coordinate one gets

```
X[t_] = Assuming[{En > 0, V > 0, a > 0, m > 0}, FS[x /. y → Y[t]]]
```

$$\frac{\text{ArcSin}\left[\frac{\text{En}\,\text{Sin}\left[\sqrt{2}\,a\,t\,\sqrt{\frac{\text{En}+V}{m}}\right]}{\sqrt{\text{En}\,(\text{En}+V)}}\right]}{a}$$

```
Plot[(X[t] /. {En → 1, a → 1, m → 1, V -> 1}), {t, 0, Pi}]
```




```
Plot[(X[t] /. {En → 10, a → 1, m → 1, V -> 1}), {t, 0, Pi}]
```




```
Plot[(X[t] /. {En → 100, a → 1, m → 1, V -> 1}), {t, 0, Pi}]
```