Mechanics 2341, PS 5

```
FS = FullSimplify;
```

Problem 1.

Integrate the equations of motion for a particle moving on the surface of a paraboloid $z = k\left(x^2 - 2x + y^2 + 4y\right)$ in a uniform gravitational field.

We first rewrite the equation of the paraboloid as
$z = k\left((x-1)^2 + (y+2)^2 - 5\right)$.

It is then clear that the motion is equivalent to the motion on the surface of the paraboloid $z = k\left(x^2 + y^2\right)$.

```
Plot3D[x^2 + y^2, {x, -2, 2}, {y, -2, 2},
 RegionFunction → Function[{x, y, z}, x^2 + y^2 ≤ 4],
 BoxRatios → Automatic, AxesLabel → {x, y, z}]
```




```
$Assumptions = {m > 0, k > 0, g > 0, En ∈ Reals};
```

Functions

```
x[t_]; y[t_]; z[t_]; r[t_]; ϕ[t_];
```

Constraint

```
z[t_] = k (x[t]^2 + y[t]^2);
```

Lagrangian

```
L = m/2 (D[x[t], t]^2 + D[y[t], t]^2 + D[z[t], t]^2) - m g z[t]
```

$$-g\,k\,m\,\left(x[t]^2+y[t]^2\right)+\frac{1}{2}m\left(x'[t]^2+y'[t]^2+k^2\left(2\,x[t]\,x'[t]+2\,y[t]\,y'[t]\right)^2\right)$$

Polar coordinates

```
x[t_] = r[t] Cos[ϕ[t]]; y[t_] = r[t] Sin[ϕ[t]];
```

Lagrangian in polar coordinates

```
L = Collect[L, {r′[t]^2, ϕ′[t]^2}, FS]
```

$$-g\,k\,m\,r[t]^2+\frac{1}{2}\left(m+4\,k^2\,m\,r[t]^2\right)r'[t]^2+\frac{1}{2}m\,r[t]^2\,\phi'[t]^2$$

Energy

```
EE = Collect[D[L, r′[t]] r′[t] + D[L, ϕ′[t]] ϕ′[t] - L, {r′[t], ϕ′[t]}, FS]
```

$$g\,k\,m\,r[t]^2+\frac{1}{2}\left(m+4\,k^2\,m\,r[t]^2\right)r'[t]^2+\frac{1}{2}m\,r[t]^2\,\phi'[t]^2$$

Angular momentum = generalized momentum of phi

```
MM = D[L, ϕ′[t]]
```

$$m\,r[t]^2\,\phi'[t]$$

Let's express r'[t] and phi'[t] as functions of En and M

```
Solve[{EE == En, MM == M}, {ϕ′[t], r′[t]}] // FS
```

$$\left\{\left\{\phi'[t]\to\frac{M}{m\,r[t]^2},\ r'[t]\to-\frac{\sqrt{-M^2+2\,m\,r[t]^2\left(En-g\,k\,m\,r[t]^2\right)}}{m\sqrt{r[t]^2+4\,k^2\,r[t]^4}}\right\},\right.$$
$$\left.\left\{\phi'[t]\to\frac{M}{m\,r[t]^2},\ r'[t]\to\frac{\sqrt{-M^2+2\,m\,r[t]^2\left(En-g\,k\,m\,r[t]^2\right)}}{m\sqrt{r[t]^2+4\,k^2\,r[t]^4}}\right\}\right\}$$

Effective potential

```
Ueff[m_, g_, k_, M_] = EE /. {ϕ′[t] → M/(m r[t]^2), r′[t] → 0} // FS
```

$$\frac{M^2}{2\,m\,r[t]^2}+g\,k\,m\,r[t]^2$$

```
Plot[(Ueff[1, 1, 1, 1] /. {r[t] → x}), {x, 0, 3}, PlotRange → {0, 10},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {x, U}}]
```



Turning points; En is the energy

```
Solve[Ueff[m, g, k, M] == En, r[t]] // FS
```

$$\left\{\left\{r[t] \to -\sqrt{\frac{M^2}{En\, m + m\sqrt{En^2 - 2\, g\, k\, M^2}}}\right\}, \left\{r[t] \to \sqrt{\frac{M^2}{En\, m + m\sqrt{En^2 - 2\, g\, k\, M^2}}}\right\}, \left\{r[t] \to -\frac{\sqrt{\frac{En+\sqrt{En^2-2\, g\, k\, M^2}}{g\, k\, m}}}{\sqrt{2}}\right\}, \left\{r[t] \to \frac{\sqrt{\frac{En+\sqrt{En^2-2\, g\, k\, M^2}}{g\, k\, m}}}{\sqrt{2}}\right\}\right\}$$

$$rmin = \sqrt{\frac{En - \sqrt{En^2 - 2\, g\, k\, M^2}}{2\, g\, k\, m}};\ rmax = \sqrt{\frac{En + \sqrt{En^2 - 2\, g\, k\, M^2}}{2\, g\, k\, m}};$$

Circle

```
Solve[rmin == rmax, M] // FS
```

... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

$$\left\{\left\{M \to -\frac{En}{\sqrt{2}\,\sqrt{g\, k}}\right\}, \left\{M \to \frac{En}{\sqrt{2}\,\sqrt{g\, k}}\right\}\right\}$$

$$r_0 = rmax\ /.\ M \to \frac{En}{\sqrt{2}\,\sqrt{g\, k}}\ //\ FS$$

$$\frac{\sqrt{\frac{En}{g\, k\, m}}}{\sqrt{2}}$$

```
DSolve[{(MM /. r[t] → r0) == M, ϕ[0] == 0}, ϕ[t], t] /. M → En/(√2 √(g k)) // FS
```

$\{\{\phi[t] \to \sqrt{2}\,\sqrt{g\,k}\;t\}\}$

```
Trev = t /. Solve[√2 √(g k) t == 2 Pi, t][[1]]
```

$$\frac{\sqrt{2}\,\pi}{\sqrt{g\,k}}$$

If rmin = rmax then the trajectory is a circle. It happens if $En^2 = 2\,g\,k\,M^2$ and the radius is

$r_0 = \sqrt{\frac{En}{2\,g\,k\,m}}$

Then, the only equation is $m\,r_0^2\,\phi'(t) = M$, and therefore $\phi(t) = t\,M/(m\,r_0^2)$. The period of revolution is

$T_{rev} = 2\pi m\,r_0^2/M = \pi\,\frac{En}{g\,k}/M = \pi\sqrt{\frac{2}{g\,k}}$. It is independent of the energy and mass.

Segment of parabola

```
rmax /. M → 0 // FS
```

$$\begin{cases} \frac{1}{\sqrt{\frac{g\,k\,m}{En}}} & En \ge 0 \\ 0 & \text{True} \end{cases}$$

If M=0 then the trajectory is a segment of a parabola, and rmin=0, $rmax = \sqrt{\frac{En}{g\,k\,m}}$. Then t as a function of r is given by $dt = dr/r'[t] = dr \Big/ \left(\frac{\sqrt{2\,r[t]\,(En - g\,k\,m\,r[t]^2)}}{\sqrt{1+4\,k^2\,r[t]^2}}\right)$

```
Assuming[{En > 0},
 FS[t = Integrate[(1/(√(2 m r[t]^2 (En - g k m r[t]^2)) / (m √(r[t]^2 + 4 k^2 r[t]^4))) /. {r[t] → r}), r]]]
```

$$\left(\sqrt{En - g\,k\,m\,r^2}\,\sqrt{r^2 + 4\,k^2\,r^4}\;\text{EllipticE}\left[\text{ArcSin}\left[\sqrt{\frac{g\,k\,m}{En}}\,r\right], -\frac{4\,En\,k}{g\,m}\right]\right) \Big/ \left(\sqrt{2}\,\sqrt{g\,k\,(1+4\,k^2\,r^2)}\,\sqrt{r^2\,(En - g\,k\,m\,r^2)}\right)$$

The period of oscillations is

```
Tosc = Assuming[{En > 0},
  FS[Integrate[(1/(Sqrt[2 m r[t]^2 (En - g k m r[t]^2)]/(m Sqrt[r[t]^2 + 4 k^2 r[t]^4]))) /. {r[t] → r}, {r, 0, Sqrt[En/(g k m)]}]]]
```

$$\frac{\text{EllipticE}\left[-\frac{4\,\text{En}\,k}{g\,m}\right]}{\sqrt{2}\,\sqrt{g\,k}}$$

```
m = 1; g = 1/2; k = Pi^2/4;
Plot[Tosc, {En, 0, 30}]
Clear[m, g, k]
```



```
Series[Tosc, {En, 0, 1}] // FS
```

$$\frac{\pi}{2\sqrt{2}\,\sqrt{g\,k}} + \frac{\sqrt{\frac{k}{g^3}}\,\pi\,\text{En}}{2\sqrt{2}\,m} + O[\text{En}]^2$$

```
Series[Tosc, {k, 0, 1}] // FS
```

$$\frac{\pi}{2\sqrt{2}\,\sqrt{g}\,\sqrt{k}} + \frac{\text{En}\,\pi\,\sqrt{k}}{2\sqrt{2}\,g^{3/2}\,m} + O[k]^{3/2}$$

```
Series[Tosc, {En, Infinity, 1}] // FS
```

$$\frac{\sqrt{2}\,\sqrt{\text{En}}}{g\,\sqrt{m}} - \frac{\left(\sqrt{m}\,\left(-1 + \text{Log}\left[\frac{g\,m}{64\,\text{En}\,k}\right]\right)\right)\sqrt{\frac{1}{\text{En}}}}{8\left(\sqrt{2}\,k\right)} + O\left[\frac{1}{\text{En}}\right]^{3/2}$$

```
Series[Tosc, {k, Infinity, 1}] // FS
```

$$\sqrt{2}\,\sqrt{\frac{\text{En}}{g^2\,m}} + \frac{m\,(1 + \text{Log}[64\,\text{En}\,k] - \text{Log}[g\,m])}{8\sqrt{2}\,\sqrt{\text{En}\,m}\,k} + O\left[\frac{1}{k}\right]^{3/2}$$

General case

In general, t as a function of r is given by $dt = dr / r'[t] = dr \Big/ \left(\frac{\sqrt{-M^2+2\,m\,r[t]^2\,(En-g\,k\,m\,r[t]^2)}}{m\,r[t]\,\sqrt{1+4\,k^2\,r[t]^2}}\right)$

```
Assuming[{k > 0, En > (2 g k)^(1/2) M, m > 0, g > 0, M > 0},
 FS[t = Integrate[(1/(Sqrt[-M^2 + 2 m r[t]^2 (En - g k m r[t]^2)]/(m Sqrt[r[t]^2 + 4 k^2 r[t]^4])) /. {r[t] → r}), r]]]
```

$$\left(i\left(2\,En\,k+g\,m-2\,k\sqrt{En^2-2\,g\,k\,M^2}\right)\sqrt{\frac{En-\sqrt{En^2-2\,g\,k\,M^2}-2\,g\,k\,m\,r^2}{2\,En\,k+g\,m-2\,k\sqrt{En^2-2\,g\,k\,M^2}}}\sqrt{\frac{En+\sqrt{En^2-2\,g\,k\,M^2}-2\,g\,k\,m\,r^2}{g\,m+2\,k\left(En+\sqrt{En^2-2\,g\,k\,M^2}\right)}}\sqrt{r^2+4\,k^2\,r^4}\left(\text{EllipticE}\left[i\,\text{ArcSinh}\left[\sqrt{-\frac{g\,m\,(1+4\,k^2\,r^2)}{g\,m+2\,k\left(En+\sqrt{En^2-2\,g\,k\,M^2}\right)}}\right],\frac{g\,m+2\,k\left(En+\sqrt{En^2-2\,g\,k\,M^2}\right)}{2\,En\,k+g\,m-2\,k\sqrt{En^2-2\,g\,k\,M^2}}\right]-\text{EllipticF}\left[i\,\text{ArcSinh}\left[\sqrt{-\frac{g\,m\,(1+4\,k^2\,r^2)}{g\,m+2\,k\left(En+\sqrt{En^2-2\,g\,k\,M^2}\right)}}\right],\frac{g\,m+2\,k\left(En+\sqrt{En^2-2\,g\,k\,M^2}\right)}{2\,En\,k+g\,m-2\,k\sqrt{En^2-2\,g\,k\,M^2}}\right]\right)\right)\Big/\left(2\,g\,k\,r\sqrt{-\frac{g\,m\,(1+4\,k^2\,r^2)}{g\,m+2\,k\left(En+\sqrt{En^2-2\,g\,k\,M^2}\right)}}\sqrt{-M^2+2\,m\,r^2\,(En-g\,k\,m\,r^2)}\right)$$

```
N[(t /. {k → 1, m → 1, g → 1, M → 1, En → 3,
    r → (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}) + 1/10^(10)}), 20] // N
```

$5.21317\times 10^{-6} + 0.286109\,i$

```
N[(t /. {k → 1, m → 1, g → 1, M → 1, En → 3,
    r → (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}) + 1/10^8}), 20] // N
```

$0.0000521317 + 0.286109\,i$

```
(t /. {k → 1, m → 1, g → 1, M → 1, En → 3.,
    r → (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}) + 0.00001})
```

0.00164857 + 0.286109 i

```
(t /. {k → 1, m → 1, g → 1, M → 1, En → 3, r → 0.6})
```

0.262601 + 0.286109 i

```
(t /. {k → 1, m → 1, g → 1, M → 1, En → 3, r → 1.})
```

0.696974 + 0.286109 i

```
(t /. {k → 1, m → 1, g → 1, M → 1, En → 3.,
    r → (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - 0.00001})
```

2.80552 + 0.286109 i

```
(t /. {k → 1, m → 1, g → 1, M → 1, En → 3.,
     r → (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - 0.00001}) - (t /. {k → 1, m → 1,
     g → 1, M → 1, En → 3., r → (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}) + 0.00001})
```

$2.80387 - 1.15274 \times 10^{-12}$ i

The imaginary part disappears in the difference t - t_0, where t_0 = t[rmin]

```
(rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3.})
(rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3.})
Plot[Re[(t /. {k → 1, m → 1, g → 1, M → 1, En → 3})],
 {r, (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}),
   (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3})},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, time}}]
```

0.420861

1.68014



```
ParametricPlot[{Re[(t /. {k → 1, m → 1, g → 1, M → 1, En → 3})], r},
 {r, (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3.}),
  (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - 1/10^6}, {AxesLabel → {time, radius}},
 PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}, PlotRange → {0, 1.8}]
```



So, phi as a function of r is given by

$$dphi = dr\, phi'[t] / r'[t] = dr \frac{M}{m\, r[t]^2} \Big/ \left(\frac{\sqrt{-M^2 + 2\, En\, m\, r[t]^2 - 2\, g\, k\, m^2\, r[t]^4}}{m\, r[t]\, \sqrt{1 + 4\, k\verb|^|2\, r[t]^2}} \right)$$

```
phi = Integrate[((M/(m r[t]^2)) / (Sqrt[-M^2 + 2 En m r[t]^2 - 2 g k m^2 r[t]^4] / (m r[t] Sqrt[1 + 4 k^2 r[t]^2])) /. {r[t] → r}), r]
```

$$\left(M \sqrt{\frac{g\, m^2 \left(1 + 4\, k^2\, r^2\right)}{2\, En\, k\, m + g\, m^2 + 2\, k \sqrt{m^2 \left(En^2 - 2\, g\, k\, M^2\right)}}} \right.$$

$$\left(4\, k \sqrt{\left(\left(En^2\, m - 2\, g\, k\, m\, M^2 - En \sqrt{m^2 \left(En^2 - 2\, g\, k\, M^2\right)} + 2\, g\, k\, m \sqrt{m^2 \left(En^2 - 2\, g\, k\, M^2\right)}\, r^2\right) \Big/ \right.}\right.$$

$$\left.\left(En^2\, m - 2\, g\, k\, m\, M^2\right)\right)$$

$$\left(- En^2\, m - En \sqrt{m^2 \left(En^2 - 2\, g\, k\, M^2\right)} + En\, g\, k\, m^2\, r^2 + g\, k\, m \left(M^2 + \sqrt{m^2 \left(En^2 - 2\, g\, k\, M^2\right)}\, r^2\right)\right)$$

$$\text{EllipticF}\left[\text{ArcSin}\left[\frac{\sqrt{\frac{\text{En}\, m+\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}-2\, g\, k\, m^2\, r^2}{\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}}}}{\sqrt{2}}\right],\right.$$

$$\left.\left(4\, k\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\right)\Big/\left(2\, \text{En}\, k\, m+g\, m^2+2\, k\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\right)\right]-$$

$$\sqrt{2}\, g\, m\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\sqrt{\frac{g\, k\left(M^2+2\, m\, r^2\left(-\text{En}+g\, k\, m\, r^2\right)\right)}{-\text{En}^2+2\, g\, k\, M^2}}$$

$$\sqrt{\left(\left(\text{En}^2\, m+\text{En}\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}-2\, g\, k\, m\left(M^2+\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\, r^2\right)\right)\Big/\right.}$$

$$\overline{\left.\left(m\left(\text{En}^2-2\, g\, k\, M^2\right)\right)\right)}$$

$$\text{EllipticPi}\left[\frac{2\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}}{\text{En}\, m+\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}},\ \text{ArcSin}\left[\frac{\sqrt{\frac{\text{En}\, m+\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}-2\, g\, k\, m^2\, r^2}{\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}}}}{\sqrt{2}}\right],\right.$$

$$\left.\left.\left.\left(4\, k\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\right)\Big/\left(2\, \text{En}\, k\, m+g\, m^2+2\, k\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\right)\right]\right)\right)\Big/$$

$$\left(g\, m\left(\text{En}\, m+\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}\right)\sqrt{1+4\, k^2\, r^2}\right.$$

$$\sqrt{\frac{\text{En}\, m+\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}-2\, g\, k\, m^2\, r^2}{\sqrt{m^2\left(\text{En}^2-2\, g\, k\, M^2\right)}}}$$

$$\left.\sqrt{-M^2+2\, m\, r^2\left(\text{En}-g\, k\, m\, r^2\right)}\right)$$

```
(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3.,
   r → (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}) + 0.00001})
```

-2.88549

```
(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3, r → 0.6})
```

-1.75297

```
(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3, r → 1.})
```

```
-1.04602
```

```
(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3.,
   r → (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - 0.00001})
```

```
-0.00312976
```

```
N[(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3,
       r → (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - 1/10^8}) -
    (phi /. {k → 1, m → 1, g → 1, M → 1, En → 3,
       r → (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3}) + 1/10^8}), 20] // N
```

```
2.8944
```

```
Plot[Re[(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3.}) - (-2.8854850582781895`)],
 {r, (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3.}) + 1/10^6,
  (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3.}) - 1/10^6},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, angle}}]
```




```
ParametricPlot[
 {Re[(phi /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - (-2.8854850582781895`)], r},
 {r, (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 3.}) + 1/10^6,
  (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 3}) - 1/10^6},
 {AxesLabel → {angle, radius}}, PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}]
```



```
phimin[k_, m_, g_, M_, En_] = N[(phi /. {r → rmin + 1/10^8}), 20];

phimin[1, 1, 1, 1, 5] // N
ParametricPlot[
 {Re[(phi /. {k → 1, m → 1, g → 1, M → 1, En → 5}) - (phimin[1, 1, 1, 1, 5])], r},
 {r, (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 5}) + 1/10^8,
  (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 5}) - 1/10^8},
 {AxesLabel → {angle, radius}}, PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}]
```

-2.75668



```
phimin[1, 1, 1, 1, 10] // N
ParametricPlot[
 {Re[(phi /. {k → 1, m → 1, g → 1, M → 1, En → 10}) - (phimin[1, 1, 1, 1, 10])], r},
 {r, (rmin /. {k → 1, m → 1, g → 1, M → 1, En → 10}) + 1 / 10^8,
  (rmax /. {k → 1, m → 1, g → 1, M → 1, En → 10}) - 1 / 10^8},
 {AxesLabel → {angle, radius}}, PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}]
```

-2.56531



Problem 2.

Consider a particle moving in the following central field

$U = \frac{\alpha - \beta r - \gamma r^2}{r^2}$

```
U = (α - β r - γ r^2)/r^2 // Expand
```

$\frac{\alpha}{r^2} - \frac{\beta}{r} - \gamma$

```
Plot[(U /. {α → 1, β → 1, γ → 1}), {r, 0.01, 10}]
```




```
F = -D[U, r]
```

$$\frac{2\,\alpha}{r^3} - \frac{\beta}{r^2}$$

```
Plot[(F /. {α → 1, β → 1, γ → 1}), {r, 0.01, 10}]
```




```
Solve[F == 0, r]
```

$$\left\{\left\{r \to \frac{2\,\alpha}{\beta}\right\}\right\}$$

```
U /. {r → 2α/β}
```

$$-\frac{\beta^2}{4\,\alpha} - \gamma$$

So, the field is repulsive for $r < \frac{2\,\alpha}{\beta}$, and attractive for $r > \frac{2\,\alpha}{\beta}$

Effective potential as a function of r and alpha, and m and M

```
Clear[Ueff]
Ueff = U + M^2/(2 m r^2)
```

$$\frac{M^2}{2\,m\,r^2}+\frac{\alpha}{r^2}-\frac{\beta}{r}-\gamma$$

Since $\alpha > 0$, the effective potential Ueff has the same shape as U with $\alpha_{\text{eff}} = \alpha + \frac{M^2}{2\,m}$, and one can have finite motion if

$-\frac{\beta^2}{4\,\alpha_{\text{eff}}} - \gamma < E < -\gamma$

and infinite motion if $E > -\gamma$

It is convenient to introduce $E_{\text{eff}} = E + \gamma$, and use α_{eff}

$$\text{Ueff} = \frac{\alpha_{\text{eff}}}{r^2} - \frac{\beta}{r} - \gamma$$

$$-\frac{\beta}{r}-\gamma+\frac{\alpha_{\text{eff}}}{r^2}$$

Finite motion: $-\frac{\beta^2}{4\,\alpha_{\text{eff}}} < E_{\text{eff}} < 0$

```
Solve[Eeff - γ - Ueff == 0, r] // FS
```

$$\left\{\left\{r\to-\frac{\beta+\sqrt{\beta^2+4\,\text{Eeff}\,\alpha_{\text{eff}}}}{2\,\text{Eeff}}\right\},\left\{r\to\frac{-\beta+\sqrt{\beta^2+4\,\text{Eeff}\,\alpha_{\text{eff}}}}{2\,\text{Eeff}}\right\}\right\}$$

$$\text{rmin} = \frac{-\beta+\sqrt{\beta^2+4\,\text{Eeff}\,\alpha_{\text{eff}}}}{2\,\text{Eeff}};\quad \text{rmax} = -\frac{\beta+\sqrt{\beta^2+4\,\text{Eeff}\,\alpha_{\text{eff}}}}{2\,\text{Eeff}};$$

Infinite motion: $E_{\text{eff}} > 0$

$$\text{rmin} = \frac{-\beta+\sqrt{\beta^2+4\,\text{Eeff}\,\alpha_{\text{eff}}}}{2\,\text{Eeff}};$$

Deflection angle

```
Clear[dr, r]
Integrand1dr = M (1/m/2)^(1/2) 1/r^2/(Eeff - γ - Ueff)^(1/2) dr
```

$$\frac{\text{dr}\sqrt{\frac{1}{m}}\,M}{\sqrt{2}\,r^2\sqrt{\text{Eeff}+\frac{\beta}{r}-\frac{\alpha_{\text{eff}}}{r^2}}}$$

$$\phi_0 = \text{Assuming}\Big[\{\text{Eeff} > 0, \alpha_{\text{eff}} > 0, \beta > 0, m > 0, M > 0\},$$
$$\text{FullSimplify}\Big[\text{Integrate}\Big[\frac{\frac{M}{\sqrt{2m}}}{r^2\sqrt{\text{Eeff} + \frac{\beta}{r} - \frac{\alpha_{\text{eff}}}{r^2}}}, \{r, \text{rmin}, \text{Infinity}\}\Big]\Big]\Big]$$

$$\left(M\left(\sqrt{2}\left(\pi + i\,\text{Log}\left[4\,\text{Eeff} + \frac{\beta^2}{\alpha_{\text{eff}}}\right]\right)\left(-\beta\sqrt{\alpha_{\text{eff}}} + \sqrt{\alpha_{\text{eff}}\left(\beta^2 + 4\,\text{Eeff}\,\alpha_{\text{eff}}\right)}\right) - 4\,i\,\text{Log}\left[2\sqrt{\text{Eeff}} + \frac{i\,\beta}{\sqrt{\alpha_{\text{eff}}}}\right]\sqrt{\left(\alpha_{\text{eff}}\left(2\,\text{Eeff}\,\alpha_{\text{eff}} + \beta\left(\beta - \sqrt{\beta^2 + 4\,\text{Eeff}\,\alpha_{\text{eff}}}\right)\right)\right)}\right)\right) \Big/ \left(4\sqrt{2}\,\alpha_{\text{eff}}\sqrt{\left(m\left(2\,\text{Eeff}\,\alpha_{\text{eff}} + \beta\left(\beta - \sqrt{\beta^2 + 4\,\text{Eeff}\,\alpha_{\text{eff}}}\right)\right)\right)}\right)$$

$$\text{Assuming}[\{\text{Eeff} > 0, \alpha_{\text{eff}} > 0, \beta > 0, m > 0, M > 0\}, \text{FS}[\text{Series}[\phi_0, \{\text{Eeff}, 0, 0\}]]]$$

$$\frac{M\pi}{\sqrt{2}\sqrt{m\,\alpha_{\text{eff}}}} - \frac{\sqrt{2}\,M\sqrt{\text{Eeff}}}{\sqrt{m}\,\beta} + O[\text{Eeff}]^1$$

$$r = 1/x;\ dr = D[r, x]\,dx$$

$$-\frac{dx}{x^2}$$

$$\text{Integrand1dr} = \text{Integrand1dr}$$

$$-\frac{dx\sqrt{\frac{1}{m}}\,M}{\sqrt{2}\sqrt{\text{Eeff} + x\,\beta - x^2\,\alpha_{\text{eff}}}}$$

$$\text{xmax} = 1/\text{rmin}$$

$$\frac{2\,\text{Eeff}}{-\beta + \sqrt{\beta^2 + 4\,\text{Eeff}\,\alpha_{\text{eff}}}}$$

$$\phi b_0 = \text{Assuming}\Big[\{\text{Eeff} > 0, \alpha_{\text{eff}} > 0, \beta > 0, m > 0, M > 0\},$$
$$\text{FullSimplify}\Big[\text{Integrate}\Big[\frac{\sqrt{\frac{1}{m}}\,M}{\sqrt{2}\sqrt{\text{Eeff} + x\,\beta - x^2\,\alpha_{\text{eff}}}}, \{x, 0, \text{xmax}\}\Big]\Big]\Big]$$

$$\left(M\left(\pi + i\,\text{Log}\left[\beta^2 + 4\,\text{Eeff}\,\alpha_{\text{eff}}\right] - 2\,i\,\text{Log}\left[i\,\beta + 2\sqrt{\text{Eeff}\,\alpha_{\text{eff}}}\right]\right)\right) \Big/ \left(2\sqrt{2}\sqrt{m\,\alpha_{\text{eff}}}\right)$$

$$\text{Assuming}[\{\text{Eeff} > 0, \alpha_{\text{eff}} > 0, \beta > 0, m > 0, M > 0\}, \text{FS}[\text{Series}[\phi b_0 - \phi_0, \{\text{Eeff}, 0, 10\}]]]$$

$$O[\text{Eeff}]^{21/2}$$

```
Integrand1dr = Integrand1dr /. x → x + β / α_eff / 2 // FS
```

$$-\frac{\sqrt{2}\ dx\ \sqrt{\frac{1}{m}}\ M}{\sqrt{4\ Eeff+\frac{\beta^2}{\alpha_{eff}}-4\ x^2\ \alpha_{eff}}}$$

```
ϕc₀ = Assuming[{Eeff > 0, α_eff > 0, β > 0, m > 0, M > 0},
  FullSimplify[Integrate[ Sqrt[2] Sqrt[1/m] M / Sqrt[4 Eeff + β^2/α_eff - 4 x^2 α_eff], {x, -β / α_eff / 2, xmax - β / α_eff / 2}]]]
```

$$\left(M\left(\pi+i\ \text{Log}\left[\beta^2+4\ Eeff\ \alpha_{eff}\right]-2\ i\ \text{Log}\left[i\ \beta+2\ \sqrt{Eeff\ \alpha_{eff}}\right]\right)\right)\Big/\left(2\ \sqrt{2}\ \sqrt{m\ \alpha_{eff}}\right)$$

```
4 Eeff + β^2/α_eff - 4 x^2 α_eff /. {x → x / 2 / α_eff^(1/2) (4 Eeff + β^2/α_eff)^(1/2)} // FS
```

$$-\frac{\left(-1+x^2\right)\ \left(\beta^2+4\ Eeff\ \alpha_{eff}\right)}{\alpha_{eff}}$$

```
ϕd₀ = Assuming[{Eeff > 0, α_eff > 0, β > 0, m > 0, M > 0},
  FullSimplify[Integrate[ Sqrt[2] M Sqrt[1/m] / Sqrt[1 - x^2] / 2 / α_eff^(1/2),
    {x, -β / α_eff / 2 2 α_eff^(1/2) / (4 Eeff + β^2/α_eff)^(1/2),
     (xmax - β / α_eff / 2) 2 α_eff^(1/2) / (4 Eeff + β^2/α_eff)^(1/2)}]]]
```

$$\frac{M\left(\pi-\text{ArcCos}\left[\frac{\beta}{\sqrt{\beta^2+4\ Eeff\ \alpha_{eff}}}\right]\right)}{\sqrt{2}\ \sqrt{m\ \alpha_{eff}}}$$

```
Assuming[{Eeff > 0, α_eff > 0, β > 0, m > 0, M > 0}, FS[Series[ϕd₀ - ϕ₀, {Eeff, 0, 10}]]]
```

$$O[Eeff]^{21/2}$$

```
ϕ₀ = Assuming[{α > 0, ρ > 0, v∞ > 0, m > 0, β > 0},
  FS[ ϕd₀ /. {Eeff → m v∞^2 / 2, α_eff → α + M^2/(2 m)} /. M → m v∞ ρ]]
```

$$\frac{m\ \rho\ \left(\pi-\text{ArcCos}\left[\frac{\beta}{\sqrt{\beta^2+2\ m\ \alpha\ v_\infty^2+m^2\ \rho^2\ v_\infty^4}}\right]\right)\ v_\infty}{\sqrt{m\ \left(2\ \alpha+m\ \rho^2\ v_\infty^2\right)}}$$

```
Assuming[{α > 0, ρ > 0, v∞ > 0, m > 0, β > 0}, FS[ϕ0 - (π - ArcCos[1/Sqrt[1+2 m α v∞^2/β^2 +m^2 ρ^2 v∞^4/β^2]])/Sqrt[1 + 2 α/(m ρ^2 v∞^2)]]]
```

0

$$\phi_0 = \frac{\left(\pi - \text{ArcCos}\left[\frac{1}{\sqrt{1+2\, m\, \alpha\, v_\infty^2/\beta^2 + m^2 \rho^2 v_\infty^4/\beta^2}}\right]\right)}{\sqrt{1 + 2\alpha/\left(m\, \rho^2\, v_\infty^2\right)}}$$

$$\frac{\pi - \text{ArcCos}\left[\frac{1}{\sqrt{1+\frac{2\, m\, \alpha\, v_\infty^2}{\beta^2}+\frac{m^2 \rho^2 v_\infty^4}{\beta^2}}}\right]}{\sqrt{1 + \frac{2\alpha}{m\, \rho^2\, v_\infty^2}}}$$

Problem 3.

Integrate the equations of motion for a particle in a central field U[r, alpha] = - alpha/r^2

```
Clear[t, phi]
```

```
$Assumptions = {alpha > 0, M > 0, m > 0, r > 0};
```

Potential as a function of r and alpha

```
U[r_, alpha_] = - alpha/r^2
```

$$-\frac{\text{alpha}}{r^2}$$

Effective potential as a function of r and alpha, and m and M

```
Ueff[r_, alpha_, m_, M_] = - alpha/r^2 + M^2/(2 m r^2)
```

$$-\frac{\text{alpha}}{r^2} + \frac{M^2}{2\, m\, r^2}$$

Case I: alpha > M^2 / (2m) and En > 0

```
Plot[{Ueff[r, 1, 1, 1], 0.6}, {r, 0, 5}, PlotRange → {-4, 1},
 PlotStyle → {{Thickness[0.008], RGBColor[0, 0, 1]},
   {Thickness[0.008], RGBColor[1, 0, 0]}}, {AxesLabel → {r, Ueff}}]
```



- Graphics -

t as a function of r

```
t = FullSimplify[(m / 2) ^ (1 / 2) Integrate[1 / (En - Ueff[r, alpha, m, M]) ^ (1 / 2), r]]
```

$$\frac{\sqrt{-M^2 + 2\,m\,(alpha + En\,r^2)}}{2\,En}$$

```
Plot[(t /. {alpha → 1, m → 1, M → 1, En → 0.6}), {r, 0, 5}, PlotRange → {0, 4},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, time}}]
```



- Graphics -

```
ParametricPlot[{(t /. {alpha → 1, m → 1, g → 1, M → 1, En → 0.6}), r},
 {r, 0, 5}, {AxesLabel → {time, radius}},
 PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}, PlotRange → {0, 6}]
```



- Graphics -

phi as a function of r

```
phi = -FullSimplify[
   M (2 m) ^ (-1 / 2) Integrate[1 / r ^ 2 × 1 / (En - Ueff[r, alpha, m, M]) ^ (1 / 2), r]]
```

$$\frac{M\,\mathrm{Log}\left[\frac{2\left(\sqrt{2\,\mathrm{alpha}\,m-M^2}+\sqrt{-M^2+2\,m\,(\mathrm{alpha}+\mathrm{En}\,r^2)}\right)}{r}\right]}{\sqrt{2\,\mathrm{alpha}\,m-M^2}}$$

```
phiInfty = Limit[phi, r → Infinity]
```

$$\frac{M\,\mathrm{Log}[8\,\mathrm{En}\,m]}{2\sqrt{2\,\mathrm{alpha}\,m-M^2}}$$

```
phi = phi - phiInfty
```

$$-\frac{M\,\mathrm{Log}[8\,\mathrm{En}\,m]}{2\sqrt{2\,\mathrm{alpha}\,m-M^2}}+\frac{M\,\mathrm{Log}\left[\frac{2\left(\sqrt{2\,\mathrm{alpha}\,m-M^2}+\sqrt{-M^2+2\,m\,(\mathrm{alpha}+\mathrm{En}\,r^2)}\right)}{r}\right]}{\sqrt{2\,\mathrm{alpha}\,m-M^2}}$$

```
Plot[(phi /. {alpha → 1, m → 1, M → 1, En → 0.6}), {r, 0, 5}, PlotRange → {-1, 6},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, angle}}]
```



- Graphics -

```
ParametricPlot[{(phi /. {alpha → 1, m → 1, g → 1, M → 1, En → 0.6}), r},
 {r, 0, 5}, {AxesLabel → {angle, radius}},
 PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}, PlotRange → {0, 6}]
```



- Graphics -

Case II: alpha > M^2 / (2m) and En < 0

```
Plot[{Ueff[r, 1, 1, 1], -0.6}, {r, 0, 5}, PlotRange → {-4, 1},
 PlotStyle → {{Thickness[0.008], RGBColor[0, 0, 1]},
   {Thickness[0.008], RGBColor[1, 0, 0]}}, {AxesLabel → {r, Ueff}}]
```



- Graphics -

The turning point

```
Assuming[En < 0, Solve[En - Ueff[r, alpha, m, M] == 0, r]]
```

$$\left\{\left\{r \to -\frac{\sqrt{-2\,\text{alpha}\,m + M^2}}{\sqrt{2}\,\sqrt{\text{En}}\,\sqrt{m}}\right\}, \left\{r \to \frac{\sqrt{-2\,\text{alpha}\,m + M^2}}{\sqrt{2}\,\sqrt{\text{En}}\,\sqrt{m}}\right\}\right\}$$

$$\text{rmax} = \frac{\sqrt{2\,\text{alpha}\,m - M^2}}{\sqrt{2}\,\sqrt{-\text{En}}\,\sqrt{m}};$$

t as a function of r

```
t = -FullSimplify[(m/2)^(1/2) Integrate[1/(En - Ueff[r, alpha, m, M])^(1/2), r]]
```

$$-\frac{\sqrt{-M^2 + 2\,m\,\left(\text{alpha} + \text{En}\,r^2\right)}}{2\,\text{En}}$$

```
t0 = Simplify[t /. {r → rmax}]
```

0

```
Plot[(t /. {alpha → 1, m → 1, M → 1, En → -0.6}),
 {r, 0, (rmax /. {alpha → 1, m → 1, M → 1, En → -0.6})}, PlotRange → {0, 1},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, time}}]
```



- Graphics -

```
ParametricPlot[{(t /. {alpha → 1, m → 1, g → 1, M → 1, En → -0.6}), r},
 {r, 0, (rmax /. {alpha → 1, m → 1, M → 1, En → -0.6})}, {AxesLabel → {time, radius}},
 PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}, PlotRange → {0, 1}]
```



- Graphics -

phi as a function of r

```
phi = -FullSimplify[
   M (2 m)^(-1/2) Integrate[1/r^2 × 1/(En - Ueff[r, alpha, m, M])^(1/2), r]]
```

$$\frac{M\,\text{Log}\left[\frac{2\left(\sqrt{2\,\text{alpha}\,m - M^2} + \sqrt{-M^2 + 2\,m\,\left(\text{alpha} + \text{En}\,r^2\right)}\right)}{r}\right]}{\sqrt{2\,\text{alpha}\,m - M^2}}$$

```
phi0 = FullSimplify[phi /. {r → rmax}]
```

$$\frac{M\,Log[-8\,En\,m]}{2\sqrt{2\,alpha\,m - M^2}}$$

```
phi = phi - phi0
```

$$-\frac{M\,Log[-8\,En\,m]}{2\sqrt{2\,alpha\,m - M^2}} + \frac{M\,Log\left[\frac{2\left(\sqrt{2\,alpha\,m - M^2} + \sqrt{-M^2 + 2\,m\,(alpha + En\,r^2)}\right)}{r}\right]}{\sqrt{2\,alpha\,m - M^2}}$$

```
Plot[(phi /. {alpha → 1, m → 1, M → 1, En → -0.6}),
 {r, 0, (rmax /. {alpha → 1, m → 1, M → 1, En → -0.6})}, PlotRange → {-1, 6},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, angle}}]
```



- Graphics -

```
ParametricPlot[{(phi /. {alpha → 1, m → 1, g → 1, M → 1, En → -0.6}), r},
 {r, 0, (rmax /. {alpha → 1, m → 1, M → 1, En → -0.6})}, {AxesLabel → {angle, radius}},
 PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]}, PlotRange → {0, 1}]
```



- Graphics -

Case III: alpha < M^2 / (2m) and En > 0

```
Plot[{Ueff[r, 1, 1, 2], 0.6}, {r, 0, 5}, PlotRange → {-1, 4},
 PlotStyle → {{Thickness[0.008], RGBColor[0, 0, 1]},
   {Thickness[0.008], RGBColor[1, 0, 0]}}, {AxesLabel → {r, Ueff}}]
```



- Graphics -

The turning point

```
Assuming[En > 0, Solve[En - Ueff[r, alpha, m, M] == 0, r]]
```

$$\left\{\left\{r \to -\frac{\sqrt{-2\,\text{alpha}\,m + M^2}}{\sqrt{2}\,\sqrt{\text{En}}\,\sqrt{m}}\right\}, \left\{r \to \frac{\sqrt{-2\,\text{alpha}\,m + M^2}}{\sqrt{2}\,\sqrt{\text{En}}\,\sqrt{m}}\right\}\right\}$$

$$\text{rmin} = \frac{\sqrt{-2\,\text{alpha}\,m + M^2}}{\sqrt{2}\,\sqrt{\text{En}}\,\sqrt{m}};$$

t as a function of r

```
t = FullSimplify[(m/2)^(1/2) Integrate[1/(En - Ueff[r, alpha, m, M])^(1/2), r]]
```

$$\frac{\sqrt{-M^2 + 2\,m\,(\text{alpha} + \text{En}\,r^2)}}{2\,\text{En}}$$

```
t0 = Simplify[t /. {r → rmin}]
```

0

```
Plot[(t /. {alpha → 1, m → 1, M → 2, En → 0.6}),
 {r, (rmin /. {alpha → 1, m → 1, M → 2, En → 0.6}), 5}, PlotRange → {0, 5},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, time}}]
```



- Graphics -

```
ParametricPlot[{(t /. {alpha → 1, m → 1, g → 1, M → 2, En → 0.6}), r},
 {r, (rmin /. {alpha → 1, m → 1, M → 2, En → 0.6}), 5},
 {AxesLabel → {time, radius}}, PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]},
 PlotRange → {(rmin /. {alpha → 1, m → 1, M → 2, En → 0.6}) - 1, 6}]
```



- Graphics -

phi as a function of r

```
phi = FullSimplify[
  M (2 m) ^ (-1 / 2) Integrate[1 / r ^ 2 × 1 / (En - Ueff[r, alpha, m, M]) ^ (1 / 2), r]]
```

$$-\frac{M\,\mathrm{Log}\left[\frac{2\left(\sqrt{2\,\mathrm{alpha}\,m-M^2}+\sqrt{-M^2+2\,m\,(\mathrm{alpha}+\mathrm{En}\,r^2)}\right)}{r}\right]}{\sqrt{2\,\mathrm{alpha}\,m-M^2}}$$

```
phi0 = FullSimplify[phi /. {r → rmin}]
```

$$-\frac{M\,\mathrm{Log}\left[\frac{2\sqrt{2}\,\sqrt{\mathrm{En}\,m}\,\sqrt{2\,\mathrm{alpha}\,m-M^2}}{\sqrt{-2\,\mathrm{alpha}\,m+M^2}}\right]}{\sqrt{2\,\mathrm{alpha}\,m-M^2}}$$

```
phi = phi - phi0
```

$$\frac{M \operatorname{Log}\left[\frac{2\sqrt{2}\sqrt{En\, m}\sqrt{2\, alpha\, m - M^2}}{\sqrt{-2\, alpha\, m + M^2}}\right]}{\sqrt{2\, alpha\, m - M^2}} - \frac{M \operatorname{Log}\left[\frac{2\left(\sqrt{2\, alpha\, m - M^2} + \sqrt{-M^2 + 2\, m\, (alpha + En\, r^2)}\right)}{r}\right]}{\sqrt{2\, alpha\, m - M^2}}$$

```
Plot[Re[(phi /. {alpha → 1, m → 1, M → 2, En → 0.6})],
 {r, (rmin /. {alpha → 1, m → 1, M → 2, En → 0.6}), 9}, PlotRange → {-1, 2.222},
 PlotStyle → {Thickness[0.008], RGBColor[0, 0, 1]}, {AxesLabel → {radius, angle}}]
```



- Graphics -

```
ParametricPlot[{(phi /. {alpha → 1, m → 1, g → 1, M → 2, En → 0.6}), r},
 {r, (rmin /. {alpha → 1, m → 1, M → 2, En → 0.6}), 9},
 {AxesLabel → {angle, radius}}, PlotStyle → {Thickness[0.008], RGBColor[1, 0, 1]},
 PlotRange → {(rmin /. {alpha → 1, m → 1, M → 2, En → 0.6}), 9}]
```



- Graphics -

phi goes to a limiting value as r -> infinity

```
Limit[phi, r → Infinity]
```

$$-\frac{M\left(\operatorname{Log}[En\, m] - 2\operatorname{Log}\left[\frac{\sqrt{En}\sqrt{m}\sqrt{2\, alpha\, m - M^2}}{\sqrt{-2\, alpha\, m + M^2}}\right]\right)}{2\sqrt{2\, alpha\, m - M^2}}$$

This expression is equal to

```
phiInfty = Pi/2 M/Sqrt[-2 alpha m + M^2]
```

$$\frac{M\,\pi}{2\sqrt{-2\,\text{alpha}\,m+M^2}}$$

It does not depend on En.

```
phiInfty /. {alpha → 1, m → 1, g → 1, M → 2}
```

$$\frac{\pi}{\sqrt{2}}$$

```
phiInfty /. {alpha → 1, m → 1, g → 1, M → 2.}
```

2.22144